<NME> build.gradle
<BEF> apply plugin: 'com.android.library'

android {
    compileSdkVersion 32
    resourcePrefix 'joda_'

    defaultConfig {
        minSdkVersion 14
        consumerProguardFiles 'proguard-rules.txt'
        testInstrumentationRunner "androidx.test.runner.AndroidJUnitRunner"
    }

    lintOptions {
        lintConfig file('lint.xml')
        textReport true
        textOutput 'stdout'
    }
}

apply plugin: net.danlew.android.joda.TzDataPlugin
apply from: 'gradle-mvn-push.gradle'

repositories {
    google()
    mavenCentral()
}

dependencies {
    api(libs.jodaTime) {
        artifact {
            classifier = "no-tzdb"
            extension = "jar"
        }
    }

    implementation libs.androidx.startup
    implementation libs.androidx.annotation

    androidTestImplementation libs.junit
    androidTestImplementation libs.androidx.test.core
    androidTestImplementation libs.androidx.test.runner
    androidTestImplementation libs.androidx.test.rules
    androidTestImplementation libs.androidx.test.ext.junit
}

tzdata {
    tzDataDir = new File(project.rootDir, 'tzdata')
}

<MSG> Switch to gradle-maven-publish-plugin

gradle-mvn-push no longer works as of Gradle 7.x, as the old Maven
publishing plugin has been removed.

<DFF> @@ -18,7 +18,7 @@ android {
 }
 
 apply plugin: net.danlew.android.joda.TzDataPlugin
-apply from: 'gradle-mvn-push.gradle'
+apply plugin: "com.vanniktech.maven.publish"
 
 repositories {
     google()

Switch to gradle-maven-publish-plugin